(* Asymptotic expected number of hairpins in ALL structures with theta=3 and stickiness p=3/8*)

(*We first compute the dominant singularity and asymptotic number of saturated structures, using Drmota–Lalley–Woods Theorem. Grammar with S, T where T means there is at least one base pair. *)

```
Clear["*"]
(* S arrow D or T; T arrow (E) or T* or (T) or S(D1) or S(T); D0 arrow * or D0 *;
D1 arrow *** or D1 *  *)
Clear[p, S, D0, D1, z, eqn0, eqn, F, z0, y0, dFdzOfz0S0, d2FdyOfz0S0];

eqn0 = {S == z + z S + z^2 S + z^2 S^2}; (* grammar for all str, theta=1,p=1 *)
CellPrint["eqn0 is the usual (simple) grammar for all sec str, with theta=1,p=1"]
eqn1 = {S == D0 + T, T == z^2 D1 + z T + z^2 T + z^2 S T + z^2 S D1,
   D0 == z + z D0, D1 == z + z D1};
CellPrint[
 "eqn1 is variant of eqn,where D1 arrow * or D1 *, giving equivalent grammar to eqn0"]
Eliminate[eqn0, {D0, D1, T}]
Collect[%, S, Simplify]
Eliminate[eqn1, {D0, D1, T}]
Collect[%, S, Simplify]
CellPrint["We derive the SAME equations
   from eqn0,eqn1 showing the underlying grammar is correct."]
p = 3 / 8;
eqn = {S == D0 + T, T == p z^2 D1 + z T + p z^2 T + p z^2 S T + p z^2 S D1,
   D0 == z + z D0, D1 == z^3 + z D1};
Eliminate[eqn, {D0, D1, T}]
Collect[%, S, Simplify]
CellPrint["From previous expression, write S = ... to get F"]
F = (3 S^2 z^2 - z (-8 + 3 z^2 + 3 z^3)) / -(-8 + 8 z + 3 z^2 - 3 z^3 - 3 z^4);

NSolve[{F == S, D[F, S] == 1}, {z, S}]
z0 = 0.539143313679122`;
y0 = 2.028866574847937`;


dFdzOfz0S0 = D[F, z] /. {z → z0, S → y0}

d2FdyOfz0S0 = D[F, {S, 2}] /. {z → z0, S → y0}
c = Sqrt[z0 dFdzOfz0S0 / (2 Pi d2FdyOfz0S0)]
c * (1 / z0)^n n^(-3 / 2)
```

eqn0 is the usual (simple) grammar for all sec str, with theta=1,p=1

eqn1 is variant of eqn,where D1 arrow * or D1 *, giving equivalent grammar to eqn0

$$S^2 z^2 + S\left(-1 + z + z^2\right) == -z$$

$$S^2 z^2 + S\left(-1 + z + z^2\right) == -z$$

$$S^2 z^2 + S\left(-1 + z + z^2\right) == -z$$

$$S^2 z^2 + S\left(-1 + z + z^2\right) == -z$$

We derive the SAME equations from eqn0,eqn1 showing the underlying grammar is correct.

$$3 S^2 z^2 + S\left(-8 + 8 z + 3 z^2 - 3 z^3 - 3 z^4\right) == z\left(-8 + 3 z^2 + 3 z^3\right)$$

$$3 S^2 z^2 + S\left(-8 + 8 z + 3 z^2 - 3 z^3 - 3 z^4\right) == z\left(-8 + 3 z^2 + 3 z^3\right)$$

From previous expression, write S = ... to get F

```
{{z → 0.553554 + 1.0139 i, S → -1.67741 + 1.24076 i},
 {z → 0.553554 - 1.0139 i, S → -1.67741 - 1.24076 i},
 {z → -1.30463 - 1.70879 i, S → -1.46093 + 0.603727 i},
 {z → -1.30463 + 1.70879 i, S → -1.46093 - 0.603727 i}, {z → 0.539143, S → 2.02887},
 {z → -1.05355 + 0.942555 i, S → -0.139082 + 0.770215 i},
 {z → -1.05355 - 0.942555 i, S → -0.139082 - 0.770215 i}, {z → 1.07012, S → 0.525995}}
```

```
8.61673
```

```
0.492886
```

```
1.22479
```

$$\frac{1.22479 \times 1.85479^n}{n^{3/2}}$$

(*Now, we compute mean,variance using Drmota's Theorem*)

```
Clear["*"]
p = 3 / 8;
eqn = {S == D0 + T, T == p u z^2 D1 + z T + p z^2 T + p z^2 S T + p u z^2 S D1,
    D0 == z + z D0, D1 == z^3 + z D1};


CellPrint["Eliminate all variables except S,u,z"]
Eliminate[eqn, {D0, D1, T}]
Collect[%, S, Simplify]
CellPrint["From previous expression, write S=... to obtain F"]
F = (3 S^2 (-1 + z) z^2 + z (-8 + 8 z + 3 z^2) - (3 u z^5 + 3 S u z^5)) / -(8 - 16 z + 5 z^2 + 6 z^3);

f = (F /. S → s)
s - f
(* express over a common denominator*)
Together[s - f]
a = Numerator[%]
c = Denominator[Together[s - f]]
dfs = D[f, s]
1 - dfs
(* express over a common denominator*)
Together[1 - dfs]
(* a is numerator of s-f, and b is numerator of 1-dfs,
where both have identical denominators *)
b = Numerator[%]
d = Denominator[Together[1 - dfs]]
(* a is numerator of s-f, and b is numerator of 1-dfs,
where both have identical denominators *)
If[Denominator[Together[s - f]] == Denominator[Together[1 - dfs]],
 CellPrint["Denominator of (s-f) same as that of (1-dfs)"],
```

```
 CellPrint["Denominator of (s-f) different than than of (1-dfs)"]]
CellPrint["Now compute the resultant of numerators a,b to get relation between S,u,z"]

(* res =Resultant[s-f,1-dfs,s] *)


(*We compute resultant of numerators,
since we have Resultant[S-F,1-D[F,S],S]=0 which is not much use.*)
CellPrint["Express S-F and 1-D[F,S] as fractions over the same common denominator"]
CellPrint["Then compute the resultant of the numerators of these expressions"]
res = Resultant[a, b, s]
(* Replace z by z[u], a function of u *)


res /. z → z[u]
(* Now compute z'[u] *)
dres = D[%, u]
Simplify[Collect[dres, z'[u]]]
Solve[dres == 0, z'[u]]
dzu = Last[Last[Last[Solve[dres == 0, z'[u]]]]];
(*z[1] equals rho, the dominant singularity *)



rho = 0.539143313679122`;
(* value of z0 in the first part of this file, the dominant singularity*)
dzuEvaluatedAt1 = (dzu /. u → 1) /. z[1] → rho
CellPrint[
 "According to Drmota's Theorem 1, the mean equals -z'[1]/z[1], computed next. "]

mu = ((-dzu / z[u] /. u → 1) /. z[1] → rho)

(*Variance computation *)
(* d2zu is z''[1] *)
d2zu = (((D[dzu, u] /. u → 1) /. z[1] → rho) /. z'[1] → dzuEvaluatedAt1)
CellPrint["Now compute variance, which by Drmota is -z''[1]/z[1] + mu^2 + mu"]
var = -d2zu / rho + mu * mu + mu
CellPrint["Now compute the standard deviation sigma"]
stdev = Sqrt[var]
```

Eliminate all variables except S,u,z

$$8\,S - 8\,z - 16\,S\,z + 8\,z^2 + 5\,S\,z^2 - 3\,S^2\,z^2 + 3\,z^3 + 6\,S\,z^3 + 3\,S^2\,z^3 == (3 + 3\,S)\,u\,z^5$$

$$3\,S^2\,(-1 + z)\,z^2 + z\left(-8 + 8\,z + 3\,z^2\right) + S\left(8 - 16\,z + 5\,z^2 + 6\,z^3\right) == 3\,u\,z^5 + 3\,S\,u\,z^5$$

From previous expression, write S=... to obtain F

$$\frac{3\,s^2\,(-1 + z)\,z^2 - 3\,u\,z^5 - 3\,s\,u\,z^5 + z\left(-8 + 8\,z + 3\,z^2\right)}{-8 + 16\,z - 5\,z^2 - 6\,z^3}$$

$$s - \frac{3\,s^2\,(-1 + z)\,z^2 - 3\,u\,z^5 - 3\,s\,u\,z^5 + z\left(-8 + 8\,z + 3\,z^2\right)}{-8 + 16\,z - 5\,z^2 - 6\,z^3}$$

$$\left(8\,s - 8\,z - 16\,s\,z + 8\,z^2 + 5\,s\,z^2 - 3\,s^2\,z^2 + 3\,z^3 + 6\,s\,z^3 + 3\,s^2\,z^3 - 3\,u\,z^5 - 3\,s\,u\,z^5\right) \Big/ \left(8 - 16\,z + 5\,z^2 + 6\,z^3\right)$$

$$8\,s - 8\,z - 16\,s\,z + 8\,z^2 + 5\,s\,z^2 - 3\,s^2\,z^2 + 3\,z^3 + 6\,s\,z^3 + 3\,s^2\,z^3 - 3\,u\,z^5 - 3\,s\,u\,z^5$$

$$8 - 16\,z + 5\,z^2 + 6\,z^3$$

$$\frac{6\,s\,(-1+z)\,z^2-3\,u\,z^5}{-8+16\,z-5\,z^2-6\,z^3}$$

$$1-\frac{6\,s\,(-1+z)\,z^2-3\,u\,z^5}{-8+16\,z-5\,z^2-6\,z^3}$$

$$\frac{8-16\,z+5\,z^2-6\,s\,z^2+6\,z^3+6\,s\,z^3-3\,u\,z^5}{8-16\,z+5\,z^2+6\,z^3}$$

$$8-16\,z+5\,z^2-6\,s\,z^2+6\,z^3+6\,s\,z^3-3\,u\,z^5$$

$$8-16\,z+5\,z^2+6\,z^3$$

Denominator of (s-f) same as that of (1-dfs)

Now compute the resultant of numerators a,b to get relation between S,u,z

Express S-F and 1-D[F,S] as fractions over the same common denominator

Then compute the resultant of the numerators of these expressions

$$192\,z^2-960\,z^3+1776\,z^4-1488\,z^5+555\,z^6-75\,z^7-144\,u\,z^7+432\,u\,z^8-486\,u\,z^9+198\,u\,z^{10}+27\,u^2\,z^{12}-27\,u^2\,z^{13}$$

$$192\,z[u]^2-960\,z[u]^3+1776\,z[u]^4-1488\,z[u]^5+555\,z[u]^6-75\,z[u]^7-144\,u\,z[u]^7+432\,u\,z[u]^8-486\,u\,z[u]^9+198\,u\,z[u]^{10}+27\,u^2\,z[u]^{12}-27\,u^2\,z[u]^{13}$$

$$-144\,z[u]^7+432\,z[u]^8-486\,z[u]^9+198\,z[u]^{10}+54\,u\,z[u]^{12}-54\,u\,z[u]^{13}+384\,z[u]\,\text{Derivative}[1][z][u]-2880\,z[u]^2\,\text{Derivative}[1][z][u]+7104\,z[u]^3\,\text{Derivative}[1][z][u]-7440\,z[u]^4\,\text{Derivative}[1][z][u]+3330\,z[u]^5\,\text{Derivative}[1][z][u]-525\,z[u]^6\,\text{Derivative}[1][z][u]-1008\,u\,z[u]^6\,\text{Derivative}[1][z][u]+3456\,u\,z[u]^7\,\text{Derivative}[1][z][u]-4374\,u\,z[u]^8\,\text{Derivative}[1][z][u]+1980\,u\,z[u]^9\,\text{Derivative}[1][z][u]+324\,u^2\,z[u]^{11}\,\text{Derivative}[1][z][u]-351\,u^2\,z[u]^{12}\,\text{Derivative}[1][z][u]$$

$$3\,z[u]\,\left(-48\,z[u]^6+144\,z[u]^7-162\,z[u]^8+66\,z[u]^9+18\,u\,z[u]^{11}-18\,u\,z[u]^{12}-\left(-128+960\,z[u]-2368\,z[u]^2+2480\,z[u]^3-1110\,z[u]^4+7\,(25+48\,u)\,z[u]^5-1152\,u\,z[u]^6+1458\,u\,z[u]^7-660\,u\,z[u]^8-108\,u^2\,z[u]^{10}+117\,u^2\,z[u]^{11}\right)\,\text{Derivative}[1][z][u]\right)$$

$$\left\{\left\{\text{Derivative}[1][z][u]\to-\left(6\,\left(8\,z[u]^6-24\,z[u]^7+27\,z[u]^8-11\,z[u]^9-3\,u\,z[u]^{11}+3\,u\,z[u]^{12}\right)\right)\Big/\left(-128+960\,z[u]-2368\,z[u]^2+2480\,z[u]^3-1110\,z[u]^4+175\,z[u]^5+336\,u\,z[u]^5-1152\,u\,z[u]^6+1458\,u\,z[u]^7-660\,u\,z[u]^8-108\,u^2\,z[u]^{10}+117\,u^2\,z[u]^{11}\right)\right\}\right\}$$

$$-0.0294578$$

According to Drmota's Theorem 1, the mean equals -z'[1]/z[1], computed next.

$$0.0546382$$

$$0.022117$$

Now compute variance, which by Drmota is -z''[1]/z[1] + mu^2 + mu

$$0.0166011$$

Now compute the standard deviation sigma

$$0.128845$$